Modeling the Mineralization and Volatilization of Nitrogen in Poultry Litter Applied to Tall Fescue

H. DON SCOTT[1]*, D. L. BAKER[1], MARK J. COCHRAN[2] and JIM SMARTT[2]

[1]Department of Agronomy, University of Arkansas, Fayetteville, AR 72701 (USA)

[2]Department of Agricultural Economics and Rural Sociology, University of Arkansas, Fayetteville, AR 72701 (USA)

ABSTRACT

Scott, H. D., Baker, D.L., Cochran, M.J., and Smartt, J. 1997. Modeling the Mineralization and Volatilization of Nitrogen in Poultry Litter Applied to Tall Fescue. Comput. Electron. Agric., 13: .

A continuous, field-scale computer model was developed to simulate the fate of N as applied in poultry litter to tall fescue. The model was designed to describe physical, chemical and biological processes of N transformation and transport within a poultry litter-forage-soil system. It is based on the solution of transient soil water flow equations simultaneously with equations describing transformation, transport, and plant uptake of various forms of N. The model was written in FORTRAN 77. The inputs required to run the model include the hydrologic characteristics of the soil profile, date and rate of application and N concentration of the litter, long-term weather, and growth habit of the forage. One particular advantage of the model is the use of a "litter" compartment in which calculations are made over time of the mineralization of organic N and volatilization of NH_3. The use of transient-state equations, which are solved with numerical techniques and adjustable time steps provide flexibility in installing other model components such as surface and water transport. A description of the subroutines that describe infiltration, runoff, redistribution and evaporation of water along with volatilization of NH_3 and plant uptake of N is given.

Keywords: animal waste management; water quality; modeling; N transport; nutrient recycling

---

*Corresponding author. email: dscott@cleora.uark.edu

INTRODUCTION

Arkansas is the leading state in the USA in broiler production with over 1.08 billion broilers produced in 1994 (Ark. Agr. Statistical Service, 1995). This large number of broilers resulted in the production of considerable quantities of waste. Historically, this broiler waste, which is known as poultry litter, has been used successfully by many growers as a fertilizer for forages such as tall fescue and bermuda grass and as a soil mulch.

Even though land disposal of poultry litter recycles nutrients back into the food production system, there is concern about potential contamination of domestic water supplies from continuous and/or heavy applications of poultry litter on agricultural soils. Reports of such contamination were cited by Liebhardt et al. (1979) and Ritter et al. (1990). Weil et al. (1990) found that under irrigated conditions, manured and non-manured fields had NO3-N concentrations of 18.3 and 15.1 mg/L, respectively. Ritter et al (1990) found that soil profile concentrations of NO3-N from plots fertilized by poultry manure were lower in the spring than in the fall suggesting greater leaching during the winter months.

Organic nitrogen in poultry litter mineralizes over time to different inorganic forms when applied to pasture, depending on the prevailing environmental conditions. From an environmental contamination perspective, the N form of most importance is NO3 which may cause serious health problems in infants due to methemoglobinemia. Extensive work has been conducted to quantify the composition and fertilizer value of poultry litter (Perkins, et al. 1964; Hileman, 1967; Giddens, et al. 1975). Several studies have characterized the mineralization of N from poultry litter under controlled laboratory conditions (Castellanos and Pratt, 1981; Hadas et al., 1983; Sims, 1986; Gale and Gilmour, 1986).

One important area that has been given little attention is the simulation of the fate of N after the application of litter to forage. Edwards et al. (1992) used the EPIC model developed by Williams et al. (1990) to study the optimum timing of poultry litter in NW Arkansas. EPIC is a continuous, field scale, lumped parameter model that simulates crop growth, runoff, soil erosion, nutrient transformation and transport. Detailed study of runoff, infiltration and redistribution of water and elements is unavailable because of the daily time step. Therefore, the objectives of this work were to: (1) develop a mechanistic, variable time step computer model that simulates the fate of nitrogenous compounds in poultry litter decomposition under differing climatic, (2) validate the model for two studies conducted in the field and (3) to apply the model of poultry litter decomposition as an input to a 1-D finite difference model of infiltration and drainage. The first two objectives are addressed here; the third will be published later.

DESCRIPTION OF THE MODEL

The model POULIT is an extensive modification of the simplified 1-D finite difference model for prediction of N behavior and transport in the land treatment of municipal effluent wastewater that was developed by Selim and Iskandar (1980) and by Ibrahim (1992). The Selim and Iskandar model was designed to simulate a weekly application of 5 cm of wastewater that contained known amounts of NH4-N and NO3-N. Our major modifications include the addition of simulation code for weather, runoff, evapotranspiration, decomposition of poultry litter, NH3-N volatilization, and soil temperature. The model described here is the subroutine for poultry litter decomposition, **litter**. Description of the full model awaits completion of the calibration and validation studies. It is used here only to produce the soil surface water content used by **litter** to estimate the litter water content.

In **litter**, the model calculates the poultry litter decomposition and nitrogen release in separate variables for each application. The input file that gives application information allows the user to specify the total amount of litter applied, **applr** (tons/acre), the percentage of that litter assigned to

nitrogen, **percen**, and the decimal fraction of the nitrogen assigned to initial inorganic nitrogen, **fci**. The main program scales the total amount of nitrogen applied in application **napp**, **tappn(napp)**, to (kg/ha), and calculates the cumulative sums and all inorganic, **csin,** and organic nitrogen, **cson**, by the formulas:

[1] tappn(i) = percen*aplr*224.265

[2] csin = csin + fci*tappn(napp)

[3] cson = cson + (1 – fci)*tappn(napp)



FIGURE 1: ORGANIC AND INORGANIC NITROGEN DECOMPOSITION

Figure 1 shows the organic and inorganic compartment for one application of litter, number **napp**. Consider modeling a litter compartment as two tanks of liquid, organic-N and inorganic-N, each starting with initial contents, **No = (1-fci)*tappn(napp)** and **Ni = fci*tappn(napp)**. **Litter** calculates amounts in non-dimensional fractions of **tappn**. The organic-N mineralizes according to the rate equation [4], where **fonr** is the fraction of organic material in **tappn** remaining in the left-had tank of Figure 1, and **akm** is the rate of mineralization to the right-hand tank. The fraction of organic nitrogen mineralized, **fam** [5], then changes with each time step by **dfam**, equation [6], from the exact integral of [4] from **fam** and time, **t** (day), to **fam + dfam** and **t + dt**.

[4] d(fonr)/dt = -akm * fonr

[5] fonr = 1 – fam(napp)

[6] dfam = fonr*(1 – exp(-akm * dt))



FIGURE 2: DECAY OF **fonr** IN EQUATION [4]

Figure 2 shows how the decay process proceeds in three phases, at 25 °C, where **akm** depends on the amount of organic matter remaining, with breakpoints in **akm** [7] at 0.8*No and 0.65*No. The model adjusts **akm** for daily average air temperature, **Ta** (°C), using the Arrhenius temperature correction factor, **tc** [8]. It corrects for moisture content [9a,b] using the soil surface water content, **th(0)**, developed from the subsurface model. The correction factors [8] and [9] are applied [10] before **akm** is used in [6]. When the organic-N tank empties to 0.15*No, the model dumps the remaining material, the residual, to the soil as unavailable-N.

$$[7] \qquad akm = \begin{Bmatrix} 0.049,\ 0.8 \le fonr \le 1 \\ 0.019,\ 0.65 \le fonr \le 0.8 \\ 0.006,\ 0 \le fonr \le 0.65 \end{Bmatrix} (1/day)$$

$$[8] \qquad tc = \begin{Bmatrix} 0,\ Ta \le 5°C \\ 0.30139\,(0.2\,(Ta + 273.15) - 1),\ 5°C \le Ta \le 10°C \\ \exp(-6750\,(1/(Ta + 273.15) - 1/298.15)),\ Ta \ge 10°C \end{Bmatrix}$$

$$[9] \qquad wcc = \begin{Bmatrix} 9\,th(0), 0 \le th(0) \le 0.1 \\ 0.8 + th(0), 0.1 \le th(0) \le 0.2 \\ 1.42735 - 2{,}13675\,th(0), 0.2 \le th(*0) \le 0.668 \end{Bmatrix}$$

[10] akm ← wcc*tc*akm

With each time step, the pool of inorganic-N in the second tank, **fam(napp)**, is increased by **dfam** and sometimes decreased by volatilization, **dfav**. Volatilization is modeled on a field experiment that considered the evolution of ammonia from the entire mass of the litter application (Scott, et al., 1995), according to [11]. The value of **dfav** is determined by integrating [11] to [12], as in [4] to [6].

[11] d(fav(napp))/dt = akv*(fmaxv – fav(napp)) , where

fav(napp) = fraction of nitrogen in application number napp volatilized,

akv = rate of volatilization (1/day), and

fmaxv = maximum fraction of applied nitrogen that can be released.

[12] dfav = (fmaxv – fav(napp)) * (1 – exp(-akv * dt))

The rate of volatilization, **akv**, and maximum volatilization, **fmaxv**, are determined by environmental factors, such as daily average air (or litter) temperature, **Ta** (°C), the accumulated rain on the particular application, **crain(napp)**, and the amount of litter nitrogen applied, **tappn(napp)**. **Litter** uses regression equations [13, 14] from an experimental calibration described latter, corrected with the same temperature coefficient, **tc**, used for mineralization. Since the regression equations can produce negative values of **dfav**, **litter** limits the result to positive values, as well as to amounts no greater than the available pool of inorganic nitrogen available to volatilize.

[13] akv/tc = -0l0170028 + 9.29486e-4*Ta – 2.41901*tappn(napp)
+ 0.00108953*crain(napp)

[14] fmaxv/tc = 0.01*(5.69783 + 0.229995*Ta +0.000459594*tappn(napp)
- 0.353627*crain(napp)

Litter also accumulates sums and daily amounts across all the applications in dimensions of ug/cm{EQ/s/up(2)}: the total mass of organic nitrogen remaining all applications of the litter, **cson**, the mass of inorganic nitrogen in the litter, **csin**, the amount mineralized per day, **ampd**, the mass volatilized per day, **avpd**, and the total cumulative amount volatilized, **cav**.

$$cson = 0,\ initially, then$$
$$cson = cson + (1 - fci) * tappn(napp),\ on\ application, then$$
[15] $$cson = cson - \sum_{i=1}^{napp} dfam(i) * tappn(i),\ on\ each\ time\ step, with$$
$$cson = cson - 0.15 * tappn(i),\ when\ application\ i\ depletes\ to\ 15\%$$

$$c\sin = 0,\ initially, then$$
[16] $$c\sin = c\sin + fci * tappn(napp),\ on\ application, then$$
$$c\sin = c\sin + \sum_{i=1}^{napp} (dfam(i) - dfav(i)) * tappn(i),\ on\ each\ time\ step$$

$$ampd = 0,\ initially,\ then$$
[17] $$ampd = ampd + \sum_{i=1}^{napp} dfam(i) * tappn(i)$$

$$avpd = 0,\ initially,\ then$$
[18] $$avpd = avpd + \sum_{i=1}^{napp} dfav(i) * tappn(i)$$

$$cav = 0,\ initially, then$$
[19] $$cav = cav + \sum_{i=1}^{napp} dfav(i) * tappn(i),\ on\ each\ time\ step$$

CALIBRATION OF THE LITTER MODEL

{Dr. Scott, please review this section and rewrite it as needed}

Scott, et al., (1995) performed an experiment with poultry litter applied to tall fescue at four different rates, 8.96, 17.92, 35.84 and 89.60 Mg/ha, at three different seasons, May 1990, November 1990 and April 1991. Using the monomolecular model implied in [11], values of **akv** and **fmaxv** were fit to the measured amounts of volatilization with regression, producing the data in Table 1. Linear regression of this data produced equations [13] and [14] with respective r-squared = 0.86456 and 0.692634, and respective standard errors of the estimates of 0.002405 and 5.255875. Figures 3 and 4 show the fit of the regression-generated volatilization parameters to the values in Table1. Correcting with the Arrhenius temperature coefficient, **tc**, in [13,14] provided slightly tighter fits at lower ranges than the regression equations presented in Scott, et al. (1995). It should be noted that the average air temperatures in Table 1 are the averages for entire 571 to 795-hour times of the field experiments, and that the cumulative rain is accumulated over those times. The subroutine **litter**

calculates **akv** and **fmaxv** from the daily average temperature and the rain accumulated from the start of the application, changing daily.

Table 1: Summary of the monomolecular model coefficients at various times and application rates

	Average Air	Cumulative	Nitrogen	Model Coefficients	
Month	Temperature	Rain	Applied	AKV	100*FMAXV
	°C	cm	kg ha	1/hr	%
May '90	22	5.7	367	.00652	4.930
			735	.00633	6.774
			1469	.00460	10.685
			3754	.00108	28.485
Nov '90			352	.00275	1.873
			704	.00193	3.094
			1409	.00112	4.783
			3521	.00089	5.552
Apr '91			375	.01033	3.818
			750	.00636	4.350
			1502	.00696	5.559
			3754	.00285	6.864



FIGURE 3: REGRESSION FIT OF [13] VERSUS ORGINAL DATA



FIGURE 4: REGRESSION FIT OF [14] VERSUS ORIGINAL DATA

LISTING 1: SUBROUTINES **LITTER** AND **ARRHEN**

```
c**********************************************************************
c* this subroutine calculates the mineralization and volatilization of
c* the litter nitrogen, and its conversion to soil OM (organic matter)
c*
c**********************************************************************

      subroutine litter


      implicit real*8(a-h,o-z)
      include 'comd8.for'

      real*8 rk(3)
c     +, ak
      data rk /0.049d0,0.019d0,0.0060d0/
c     +,  ak /-6750.d0/

      tc = arrhen (-6750.d0, tave)
      do i = 1, napp
         if (fam(i) .lt. 0.85d0) then

c*------- calculate volatilization if any inorganic-N left

            dfav = 0.d0
            if (fav(i) .lt. fam(i)) then
               fmaxv = 0.01d0*( 5.69783d0 +0.229995d0*tave
     +                  +0.000459594d0*tappn(i) -0.353627d0*crain(i) )
               akv = -0.0170028d0 +9.29487d-4*tave -2.41901d-7*tappn(i)
     +               +0.00108953d0*crain(i)
               if ((akv .gt. 0.d0) .and. (fmaxv .gt. 0.d0)) then
                  akv = tc*akv
                  fmaxv = tc*fmaxv
                  dfav = (fmaxv - fav(i))*(1.d0 - dexp(-akv*dt))
                  if (dfav .lt. 0.d0) dfav = 0.d0
                  if ((fav(i)+dfav) .ge. fam(i)) dfav = fam(i) -fav(i)
                  fav(i) = fav(i) + dfav
               end if
            end if

            d3 = dfav*tappn(i)
            if (d3 .gt. csin) d3 = csin
            avpd = avpd + d3
            cav = cav + d3
            csin = csin - d3

c*------- the mineralization process rate constant

            fonr = 1.d0 - fam(i)
            if ((1.d0-fam(i)) .gt. 0.8d0) then
               akm = rk(1)
            else if ((1.d0-fam(i)) .gt. 0.65d0) then
               akm = rk(2)
            else
               akm = rk(3)
            end if
            akm = akm/24.d0

c*------- water content correction

            if (th(0) .lt. 0.1d0) then
               ae=9.d0
               af=0.d0
            elseif (th(0) .lt. 0.2d0) then
```

```
               ae=1.d0
               af=0.8d0
            else if (th(0) .lt. 0.668d0) then
               ae=-2.13675d0
               af=1.42735d0
            else
               ae = 0.d0
               af = 0.d0
            endif

            akm=akm*(ae*th(0)+af)

c*------- air temp adjustment to bacterial action

            akm = tc*akm

c*------- mineralization

            dfam = fonr*(1.d0 - dexp(-akm*dt))
            if ((fam(i)+dfam) .ge. 0.85d0) dfam = 0.85d0 - fam(i)
            fam(i) = fam(i) + dfam
            d3 = dfam*tappn(i)
            csin = csin + d3
            ampd = ampd + d3
            cson = cson - d3
         else
            go to 10
         end if

c*------- if only 15% of total applied N left, dump remainder to soil
c*        organic matter, and reset litter to zero

         if (fam(i) .ge. 0.85d0) then
            don = (1.d0 - fam(i))*tappn(i)
            cson = cson - don
            csom = csom + don
         end if

      end do

   10 continue
      cslr = csin + cson

      return

      end


c*-------
c This function calculates the Arrhenius temperature correction
c to a chemical rate involving bacterial action
c    a = "activation energy" at 25 deg-C (deg-Kelvin)
c    t = current temperature of the reaction (deg-C)
c*-------

      function arrhen (a, tc)

      implicit real*8(a-h,o-z)

      t = tc + 273.15d0
      k10 = exp(a*15.d0/(283.15d0*298.15d0))
      if (tc .le. 5.d0) then
         arrhen = 0.d0
      else if (tc .lt. 10.d0) then
         arrhen = k10*(0.2d0*tc - 1.d0)
      else
         arrhen = exp(a*(1/t - 1/298.15))
```

```
end if

return
end
```